## DAO (Demostración Asistida por Ordenador) con Lean

José A. Alonso Jiménez

Grupo de Lógica Computacional Dpto. de Ciencias de la Computación e Inteligencia Artificial Universidad de Sevilla

Sevilla, 23 de agosto de 2020

Esta obra está bajo una licencia Reconocimiento-NoComercial-Compartirlgual 2.5 Spain de Creative Commons.

#### Se permite:

- copiar, distribuir y comunicar públicamente la obra
- hacer obras derivadas

#### **Bajo las condiciones siguientes:**



**Reconocimiento**. Debe reconocer los créditos de la obra de la manera especificada por el autor.



**No comercial**. No puede utilizar esta obra para fines comerciales.



**Compartir bajo la misma licencia**. Si altera o transforma esta obra, o genera una obra derivada, sólo puede distribuir la obra generada bajo una licencia idéntica a ésta.

- Al reutilizar o distribuir la obra, tiene que dejar bien claro los términos de la licencia de esta obra.
- Alguna de estas condiciones puede no aplicarse si se obtiene el permiso del titular de los derechos de autor.

Esto es un resumen del texto legal (la licencia completa). Para ver una copia de esta licencia, visite <a href="http://creativecommons.org/licenses/by-nc-sa/2">http://creativecommons.org/licenses/by-nc-sa/2</a>. 5/es/ o envie una carta a Creative Commons, 559 Nathan Abbott Way, Stanford, California 94305, USA.

## **Índice general**

1	Introducción	5
2	Igualdad2.1 Prueba mediante reescritura2.2 Prueba con lemas y mediante encadenamiento de ecuaciones2.3 Teorema aritmético con hipótesis y uso de lemas2.4 Ejercicios sobre aritmética real	11 13
3	Conectivas: implicación, equivalencia, conjunción y disyunción)	23
•	3.1 Reglas de la implicación	_
	3.1.1 Eliminación de la implicación	
	3.1.2 Introducción de la implicación	
	3.2 Reglas de la equivalencia	
	3.2.1 Eliminación de la equivalencia	
	3.3 Reglas de la conjunción	
	3.3.1 Eliminación de la conjunción	
	3.3.2 Introducción de la conjunción	
	3.4 Reglas de la disyunción	
	3.4.1 Eliminación de la disyunción	39
	3.5 Ejercicios	41
	3.5.1 Monotonía de la suma por la izquierda	41
	3.5.2 Monotonía de la suma por la derecha	
	3.5.3 La suma de no negativos es expansiva	
	3.5.4 Suma de no negativos	
	3.5.5 Suma de desigualdades	
	3.5.6 Monotonía de la multiplicación por no negativo	54
4	Apéndices	57
•	4.1 Resumen de tácticas usadas	
	4.1.1 Demostraciones estructuradas	
	4.1.2 Composiciones y descomposiciones	

	,
Λ	Índice general
4	indice general
•	maice general

4.2	Resumen de teoremas usados											60
4.3	Estilos de demostración											61

## Capítulo 1

## Introducción

El objetivo de este trabajo es presentar una introducción a la DAO (Demostración Asistida con Ordenador) usando Lean para usarla en las clases de la asignatura de Razonamiento automático del Máster Universitario en Lógica, Computación e Inteligencia Artificial de la Universidad de Sevilla. Por tanto, el único prerrequisito es, como en el Máster, cierta madurez matemática como la que deben tener los alumnos de los Grados de Matemática y de Informática.

La exposición se hará mediante una colección de ejercicios. En cada ejercicios se mostrarán distintas pruebas del mismo resultado y se comentan las tácticas conforme se van usando y los lemas utilizados en las demostraciones.

Además, en cada ejercicio hay tres enlaces: uno al código, otro que al pulsarlo abre el ejercicio en Lean Web (en una sesión del navegador) de forma que se puede navegar por las pruebas y editar otras alternativas, y el tercero es un enlace a un vídeo explicando las soluciones del ejercicio.

El trabajo se presenta en 2 formas:

- Como un libro en PDF
- Como un proyecto en GitHub.

Además, los vídeos correspondientes a cada uno de los ejercicios se encuentran en YouTube.

El trabajo se basa fundamentalmente en el proyecto lean-tutorials de la Comunidad Lean que, a su vez, se basa en el curso Introduction aux mathématiquesformalisées de Patrick Massot.

## Capítulo 2

## Igualdad

En este capítulos se presenta el razonamiento con igualdades mediante reescritura.

#### 2.1. Prueba mediante reescritura

```
example
 (h : x = y)
 (h' : y = z)
 : x = z :=
begin
 rw h,
 exact h',
end
-- Prueba:
/-
 x y z : \mathbb{R},
 h: x = y
 h': y = z
 \vdash x = z
rw h,
 \vdash y = z
exact h',
 no goals
-/
-- Comentarios:
-- + La táctica (rw h) cuando h es una igualdad sustituye en la
-- conclusión el término izquierdo de h por el derecho.
-- + La táctica (exact h) concluye la demostración si h es del tipo de
-- la conclusión.
-- 2ª demostración (con reescritura inversa)
example
 (h : x = y)
 (h': y = z)
 : x = z :=
begin
  rw ← h',
 exact h,
end
-- Prueba:
 x y z : \mathbb{R},
 h: x = y,
 h': y = z
 \vdash x = z
```

```
rw \leftarrow h',
 \vdash x = y
exact h,
 no goals
-/
-- Comentarios:
-- + La táctica (rw ← h) cuando h es una igualdad sustituye en la
-- conclusión el término derecho de h por el izquierdo
-- 3ª demostración (con reescritura en hipótesis)
-- -----
example
 (h : x = y)
 (h': y = z)
 : X = Z :=
begin
 rw h' at h,
 exact h,
end
-- Prueba:
 x y z : \mathbb{R},
 h: x = y,
 h': y = z
 \vdash x = z
rw h' at h,
 h: x = z
 \vdash x = z
exact h,
 no goals
-/
-- Comentarios:
-- + La táctica (rw h1 at h2) cuando h1 es una igualdad sustituye en la
    hipótesis h2 el término izquierdo de h1 por el derecho.
-- 4ª demostración (con reescritura inversa en hipótesis)
example
 (h : x = y)
 (h': y = z)
```

```
: X = Z :=
begin
  rw ← h at h',
  exact h',
end
-- Prueba:
 x y z : \mathbb{R},
 h: x = y,
 h': y = z
 \vdash x = z
rw \leftarrow h \ at \ h',
 h': X = Z
  \vdash x = z
exact h',
 no goals
-/
-- Comentarios:
-- + La táctica (rw ← h1 at h2) cuando h1 es una igualdad sustituye en la
-- hipótesis h2 el término derecho de h1 por el izquierdo
-- 5ª demostración (con un lema)
example
 (h : x = y)
  (h': y = z)
  : x = z :=
eq.trans h h'
-- Comentarios:
-- + Se ha usado el lema
-- + eq.trans : a = b \rightarrow b = c \rightarrow a = c
-- + El lema se puede encontrar con
      by suggest
-- 6ª demostración (por sustitución)
-- -----
example
  (h : x = y)
  (h' : y = z)
  : x = z :=
```

```
h' ▶ h
-- Comentario:
-- + Si h es una igualdad entonces h ▶ h' es la expresión obtenida sustituyendo
-- en h' el término izquierdo de h por el derecho.
-- 7ª demostración (automática con linarith)
-- -----
example
 (h : x = y)
 (h' : y = z)
 : X = Z :=
by linarith
-- Comentarios:
-- + La táctica linarith demuestra la conclusión mediante aritmética
-- + La sugerencia de usar linarith se puede obtener escrbiendo
-- by hint
-- 8ª demostración (automática con finish)
-- -----
example
 (h : x = y)
 (h': y = z)
 : X = Z :=
by finish
-- Comentario:
-- + La táctica finish demuestra la conclusión de forma automática.
-- + La sugerencia de usar finish se puede obtener escrbiendo
-- by hint
```

## 2.2. Prueba con lemas y mediante encadenamiento de ecuaciones

```
-- En esta relación se presentan distintas pruebas con Lean de una
-- igualdad con productos de números reales. La primera es por
-- reescritura usando las propiedades asociativa y conmutativa, La
-- segunda es con encadenamiento de ecuaciones. Las restantes son
-- automáticas.
-- Ejercicio. Sean a, b y c números reales. Demostrar que
-- (a * b) * c = b * (a * c)
import data.real.basic
variables (a b c : R)
-- 1º demostración (hacia atrás con rw)
example : (a * b) * c = b * (a * c) :=
begin
 rw mul_comm a b,
  rw mul_assoc,
end
-- Prueba:
/-
 abc:\mathbb{R}
 \vdash (a * b) * c = b * (a * c)
rw mul comm a b,
\vdash (b * a) * c = b * (a * c)
rw mul_assoc,
 no goals
-/
-- Comentarios:
-- + Se han usado los lemas
-- + mul comm : \forall (a b : \mathbb{R}), a * b = b * a
-- + mul assoc : \forall (a b c : \mathbb{R}), a * b * c = a * (b * c)
-- 2ª demostración (encadenamiento de igualdades)
example : (a * b) * c = b * (a * c) :=
begin
calc (a * b) * c = (b * a) * c : by rw mul_comm a b
```

## 2.3. Teorema aritmético con hipótesis y uso de lemas

```
-- En esta relación comentan distintas pruebas con Lean de una igualdad
-- con productos de números reales. La primera es por reescritura usando
-- las propiedades asociativa y conmutativa, La segunda es con
-- encadenamiento de ecuaciones. Las restantes son automáticas.

-- Ejercicio. Sean a, b, c y d números reales. Demostrar que si
-- c = d * a + b
-- b = a + d
-- entonces c = 2 * a * d.
```

```
import data.real.basic
variables (a b c d : \mathbb{R})
-- 1º demostración (reescribiendo las hipótesis)
example
  (h1 : c = d * a + b)
  (h2 : b = a * d)
  : c = 2 * a * d :=
begin
  rw h2 at h1,
  rw mul comm at h1,
  rw \leftarrow two_mul (a * d) at h1,
  rw ← mul_assoc at h1,
  exact h1,
end
-- Prueba:
 abcd: \mathbb{R},
 h1 : c = d * a + b,
 h2 : b = a * d
 \vdash c = 2 * a * d
rw h2 at h1,
 h1 : c = d * a + a * d
  \vdash c = 2 * a * d
rw mul comm at h1,
 h1 : c = a * d + a * d
  \vdash c = 2 * a * d
rw \leftarrow two_mul(a * d) at h1,
 h1 : c = 2 * (a * d)
  + c = 2 * a * d
rw ← mul assoc at h1,
 h1 : c = 2 * a * d
 \vdash c = 2 * a * d
exact h1,
 no goals
-/
-- Comentarios:
-- + Se han usado los siguientes lemas
-- + mul\_comm : \forall (a b : \mathbb{R}), a * b = b * a
```

```
-- + mul assoc : \forall (a b c : \mathbb{R}), a * b * c = a * (b * c)
-- + two_mul : 2 * a = a + a
-- 2ª demostración (encadenamiento de ecuaciones)
example
 (h1 : c = d * a + b)
 (h2 : b = a * d)
 : c = 2 * a * d :=
begin
 calc
   c = d * a + b : by exact h1
    ... = d * a + a * d : by rw h2
   \dots = a * d + a * d : by rw mul_comm
   ... = 2 * (a * d) : by rw two_mul (a * d)
... = 2 * a * d : by rw mul_assoc,
end
-- 3ª demostración (encadenamiento de ecuaciones)
example
 (h1 : c = d * a + b)
  (h2 : b = a * d)
 : c = 2 * a * d :=
begin
 calc
   c = d * a + b : by exact h1
   ... = d * a + a * d : by rw h2
    \dots = 2 * a * d : by ring,
end
-- 4ª demostración (automática con linarith)
example
 (h1 : c = d * a + b)
 (h2 : b = a * d)
 : c = 2 * a * d :=
by linarith
```

## 2.4. Ejercicios sobre aritmética real

```
-- En esta relación se comentan distintas pruebas con Lean de ejercicios
-- sobre la aritmética de los números reales. La primera es por
-- reescritura, la segunda es con encadenamiento de ecuaciones y las
-- restantes son automáticas.
-- Ejercicio 1. Ejecutar las siguientes acciones:
-- 1. Importar la teoría de los números reales.
-- 2. Declarar a, b, c y d como variables sobre los reales.
import data.real.basic -- 1
variables (a b c d : \mathbb{R}) -- 2
-- Ejercicio 2. Demostrar que
-- (c * b) * a = b * (a * c)
-- Indicación: Para alguna pueba pueden ser útiles los lemas
-- + mul_assoc : (a * b) * c = a * (b * c)
-- + mul comm : a * b = b * a
-- ---------
-- 1ª demostración
-- ==========
example : (c * b) * a = b * (a * c) :=
begin
  rw mul_comm c b,
 rw mul_assoc,
 rw mul_comm c a,
end
-- Prueba:
/-
 abc:\mathbb{R}
 \vdash (c * b) * a = b * (a * c)
rw mul comm c b,
 \vdash (b * c) * a = b * (a * c)
rw mul assoc,
```

```
\vdash b * (c * a) = b * (a * c)
rw mul_comm c a,
 no goals
-/
-- 2ª demostración
-- ==========
example : (c * b) * a = b * (a * c) :=
begin
 calc (c * b) * a = (b * c) * a : by rw mul_comm c b
              \dots = b * (c * a) : by rw mul_assoc
               \dots = b * (a * c) : by rw mul_comm c a,
end
-- 3ª demostración
-- ==========
example : (c * b) * a = b * (a * c) :=
by linarith
-- 4ª demostración
-- ==========
example : (c * b) * a = b * (a * c) :=
by finish
-- 5ª demostración
-- ===========
example : (c * b) * a = b * (a * c) :=
by ring
-- Ejercicio 3. Demostrar que si
c = b * a - d
d = a * b
-- entonces c = 0.
-- Indicación: Para alguna pueba pueden ser útiles los lemas
-- + mul\_comm : a * b = b * a
-- + sub self : a - a = 0
example
```

```
(h1 : c = b * a - d)
  (h2 : d = a * b)
  : c = 0 :=
begin
  rw h2 at h1,
 rw mul_comm b a at h1,
 rw sub_self (a * b) at h1,
 exact h1,
end
-- Prueba:
 abcd:\mathbb{R},
 h1: c = b * a - d,
 h2: d = a * b
 \vdash c = 0
rw h2 at h1,
 h1 : c = b * a - a * b
 \vdash c = 0
rw mul comm b a at h1,
 h1 : c = a * b - a * b
 \vdash c = 0
rw sub_self (a * b) at h1,
 h1 : c = 0
 \vdash c = 0
exact h1,
 no goals
-/
-- 2ª demostración
-- ==========
example
 (h1 : c = b * a - d)
 (h2 : d = a * b)
 : c = 0 :=
begin
 calc c = b * a - d : by rw h1
    ... = b * a - a * b : by rw h2
     \dots = a * b - a * b : by rw mul_comm a b
               : by rw sub_self (a*b),
end
-- 3ª demostración
-- ==========
```

```
example
 (h1 : c = b * a - d)
 (h2 : d = a * b)
 : c = 0 :=
begin
 calc c = b * a - d : by rw h1
     ... = b * a - a * b : by rw h2
    ... = 0
               : by ring,
end
-- Ejercicio 4. Demostrar que
-- (a + b) + a = 2 * a + b
-- Indicación: Para alguna pueba pueden ser útiles los lemas
-- + add \ assoc : (a + b) + c = a + (b + c)
-- + add\_comm : a + b = b + a
-- + two mul : 2 * a = a + a
-- 1ª demostración
-- ==========
example: (a + b) + a = 2 * a + b :=
 calc (a + b) + a = a + (b + a): by rw add assoc
              \dots = a + (a + b) : by rw add_comm b a
              \dots = (a + a) + b : by rw \leftarrow add_assoc
              \dots = 2 * a + b : by rw two_mul,
end
-- 2ª demostración
-- ===========
example: (a + b) + a = 2 * a + b :=
by ring
-- Ejercicio 5. Demostrar que
-- (a + b) * (a - b) = a^2 - b^2
-- Indicación: Para alguna pueba pueden ser útiles los lemas
-- + add_mul : (a + b) * c = a * c + b * c
-- + add \ sub : a + (b - c) = (a + b) - c
```

```
-- + add zero : a + 0 = a
-- + mul_comm : a * b = b * a
-- + mul\_sub : a * (b - c) = a * b - a * c
-- + pow_two : a^2 = a * a
-- + sub_self : a - a = 0
-- + sub sub : (a - b) - c = a - (b + c)
-- 1ª demostración
- - ===========
example : (a + b) * (a - b) = a^2 - b^2 :=
begin
  rw pow_two a,
  rw pow_two b,
  rw mul_sub (a + b) a b,
  rw add mul a b a,
  rw add mul a b b,
  rw mul comm b a,
  rw ← sub sub,
  rw ← add_sub,
  rw sub_self,
  rw add_zero,
end
-- Prueba:
/-
 ab:\mathbb{R}
 \vdash (a + b) * (a - b) = a ^ 2 - b ^ 2
rw pow two a,
 \vdash (a + b) * (a - b) = a * a - b ^ 2
rw pow two b,
  \vdash (a + b) * (a - b) = a * a - b * b
rw mul sub (a + b) a b,
 \vdash (a + b) * a - (a + b) * b = a * a - b * b
rw add mul a b a,
 \vdash a * a + b * a - (a + b) * b = a * a - b * b
rw add mul a b b,
 \vdash a * a + b * a - (a * b + b * b) = a * a - b * b
rw mul comm b a,
 \vdash a * a + a * b - (a * b + b * b) = a * a - b * b
rw ← sub sub,
  \vdash a * a + a * b - a * b - b * b = a * a - b * b
rw ← add sub,
 \vdash a * a + (a * b - a * b) - b * b = a * a - b * b
```

```
rw sub self,
 \vdash a * a + 0 - b * b = a * a - b * b
rw add zero,
 no goals
-/
-- 2ª demostración
-- ===========
example : (a + b) * (a - b) = a^2 - b^2 :=
begin
 calc (a + b) * (a - b)
       = (a + b) * a - (a + b) * b : by rw mul_sub (a + b) a b ... = a * a + b * a - (a + b) * b : by rw add_mul a b a
       \dots = a * a + b * a - (a * b + b * b) : by rw add_mul a b b
       \dots = a * a + a * b - (a * b + b * b) : by rw mul_comm b a
       \dots = a * a + (a * b - a * b) - b * b : by rw add sub
       \dots = a * a + 0 - b * b
                                            : by rw sub self
       ... = a * a - b * b
                                            : by rw add zero
       ... = a^2 - b * b
                                            : by rw pow two a
       ... = a^2 - b^2
                                             : by rw pow two b,
end
-- 3ª demostración
-- ===========
example : (a + b) * (a - b) = a^2 - b^2 :=
by ring
```

## Capítulo 3

# Conectivas: implicación, equivalencia, conjunción y disyunción)

## 3.1. Reglas de la implicación

#### 3.1.1. Eliminación de la implicación

```
example
 (h1 : P \rightarrow Q)
  (h2 : P)
  : Q :=
begin
 apply h1,
 exact h2,
end
-- Prueba:
 P Q : Prop,
 h1: P \rightarrow Q,
 h2 : P
 ⊢ Q
apply h1,
 \vdash P
exact h2,
 no goals
-- Comentarios:
-- + La táctica (apply h), cuando h es una implicación, aplica la regla
    de eliminación de la implicación; es decir, si h es (P \rightarrow Q) y la
    conclusión coincide con Q, entonces sustituye la conclusión por P.
-- 2ª demostración (hacia adelante)
-- -----
example
 (h1 : P \rightarrow Q)
 (h2 : P)
 : Q :=
begin
 exact h1 h2,
end
-- Comentarios:
-- + Si h1 es una demostración de (P \rightarrow Q) y h2 es una demostración de P,
-- entonces (h1 h2) es una demostración de Q.
-- 3ª demostración (simplificació de la 2ª)
--
```

```
example
 (h1 : P \rightarrow Q)
 (h2 : P)
 : Q :=
by exact h1 h2
-- 4ª demostración (mediante un término)
--
example
 (h1 : P \rightarrow Q)
 (h2 : P)
 : Q :=
h1 h2
-- 5ª demostración (automática con tauto)
example
 (h1 : P \rightarrow Q)
 (h2 : P)
 : Q :=
by tauto
-- Comentarios:
-- + La táctica tauto demuestra automáticamente las tautologías.
-- 6ª demostración (automática con finish)
-- -----
example
 (h1 : P \rightarrow Q)
 (h2 : P)
 : Q :=
by finish
-- 6ª demostración (automática con solve by elim)
-- ------
example
 (h1 : P \rightarrow Q)
 (h2 : P)
 : Q :=
by solve_by_elim
```

```
-- Comentarios:
-- + La táctica solve_by_elim intnta demostrar el objetivo aplicándole
-- reglas de eliminación.
```

## 3.1.2. Introducción de la implicación

```
-- En este relación se muestra distintas formas de demostrar un teorema
-- con eliminación de la implicación.
-- Ejercicio. Realizar las siguientes acciones:
-- 1. Importar la librería de tácticas.
-- 2. Declarar P como variable sobre proposiciones.
import tactic
variables (P : Prop) -- 2
-- Ejercicio. Demostrar que
-- P → P
-- 1ª demostración
-- ==========
example : P \rightarrow P :=
begin
 intro h,
 exact h,
end
-- Prueba:
 P : Prop
 \vdash P \rightarrow P
intro h,
 h : P
 \vdash P
exact h,
 no goals
```

```
-/
-- Comentarios:
-- + La táctica (intro h), cuando la conclusión es una implicación,
    aplica la regla de introducción de la implicación; es decir, si la
    conclusión es (P \rightarrow Q) entonces añade la hipótesis (h : P) y cambia
-- la conclusión a Q.
-- 3ª demostración (por un término)
example : P → P :=
λh, h
-- 4ª demostración (mediante id)
example : P \rightarrow P :=
id
-- Comentario: Se usa el lema
-- + id : P \rightarrow P
-- 5ª demostración (estructurada)
example : P \rightarrow P :=
begin
 assume h : P,
 show P, from h,
end
-- 6ª demostración (estructurada)
example : P \rightarrow P :=
assume h, h
-- 7º demostración (automática con tauto)
example : P \rightarrow P :=
by tauto
-- 8ª demostración (automática con finish)
```

## 3.2. Reglas de la equivalencia

## 3.2.1. Eliminación de la equivalencia

```
example
  (h : P \leftrightarrow Q)
  (hQR : Q \rightarrow R)
  : P \rightarrow R :=
begin
  intro hP,
  apply hQR,
  cases h with hPQ hQP,
  apply hPQ,
  exact hP,
end
-- Prueba:
 P Q R : Prop,
 h:P\leftrightarrow Q,
  hQR : Q \rightarrow R
  \vdash P \rightarrow R
intro hP,
  hP : P
  \vdash R
apply hQR,
  ⊢ Q
cases h with hPQ hQP,
  hPQ: P \rightarrow Q
  hQP : Q \rightarrow P
  ⊢ Q
apply hPQ,
  \vdash P
exact hP,
 no goals
-/
-- Comentarios:
-- + La táctica (cases h with h1 h2), cuando la hipótesis h es una
      equivalencia aplica la regla de eliminación de la equivalencia; es
      decir, si h es (P \leftrightarrow Q), entonces elimina h y añade las hipótesis
-- (h1 : P \to Q) \ y \ (h2 : Q \to P).
-- 2ª demostración (simplificando los últimos pasos de la anterior)
example
  (h : P \leftrightarrow Q)
```

```
(hQR : Q \rightarrow R)
  : P \rightarrow R :=
begin
 intro hP,
  apply hQR,
 cases h with hPQ hQP,
  exact hPQ hP,
end
-- Prueba:
/-
 P Q R : Prop,
 h: P \leftrightarrow Q
 hQR : Q \rightarrow R
  \vdash P \rightarrow R
intro hP,
 hP : P
  \vdash R
apply hQR,
  ⊢ Q
cases h with hPQ hQP,
 hPQ: P \rightarrow Q,
 hQP : Q \rightarrow P
 \vdash Q
exact hPQ hP,
 no goals
-/
-- 3ª demostración (simplificando los últimos pasos de la anterior)
example
 (h : P \leftrightarrow Q)
  (hQR : Q \rightarrow R)
 : P → R :=
begin
 intro hP,
  exact hQR (h.1 hP),
end
-- Comentarios:
-- + Si h es la equivalencia (P ↔ Q), entonces h.1 es (P → Q) y h.2 es
-- (Q \rightarrow P).
-- 4º demostración (por un término)
```

```
example
 (h : P \leftrightarrow Q)
  (hQR : Q \rightarrow R)
  : P → R :=
\lambda hP, hQR (h.1 hP)
-- 5ª demostración (por reescritura)
example
 (h : P \leftrightarrow Q)
  (hQR : Q \rightarrow R)
  : P → R :=
begin
 rw h,
 exact hQR,
end
-- Prueba:
/-
 P Q R : Prop,
 h:P\leftrightarrow Q,
 hQR : Q \rightarrow R
 \vdash P \rightarrow R
rw h,
  \vdash Q \rightarrow R
exact hQR,
 no goals
-/
-- Comentarios:
-- + La táctica (rw h), cuando h es una equivalencia como (P ↔ Q),
     sustituye en la conclusión P por Q.
-- 6ª demostración (por reescritura en hipótesis)
example
 (h : P \leftrightarrow Q)
  (hQR : Q \rightarrow R)
  : P → R :=
begin
  rw \leftarrow h at hQR,
```

```
exact hQR,
end
-- Prueba:
 P Q R : Prop,
 h: P \leftrightarrow Q,
 hQR : Q \rightarrow R
 \vdash P \rightarrow R
rw \leftarrow h at hQR,
 hQR : P \rightarrow R
 \vdash P \rightarrow R
exact hQR,
 no goals
-/
-- Comentarios:
-- + La táctica (rw ← h at h'), cuando h es una equivalencia como (P ↔- Q),
-- sustituye en la hipótesis h' la fórmula Q por P.
-- 7º demostración (estructurada)
example
 (h : P \leftrightarrow Q)
  (hQR : Q \rightarrow R)
  : P → R :=
begin
  assume hP : P,
 have hQ : Q, from h.1 hP,
 show R, from hQR hQ,
end
-- Comentarios:
-- + La táctica (assume h : P), cuando la conclusión es de la forma
-- (P → Q), añade la hipótesis P y cambia la conclusión a Q.
-- + La táctica (have h : e) genera dos subojetivos: el primero tiene
-- como conclusión e y el segundo tiene la conclusión actual pero se le
     añade la hipótesis (h : e).
-- + la táctica (show P, from h) demuestra la conclusión con la prueba h.
-- 8ª demostración (estructurada)
example
```

## 3.3. Reglas de la conjunción

## 3.3.1. Eliminación de la conjunción

```
P Q : Prop
  \vdash P \land Q \rightarrow P
intro h,
  h: P \wedge Q
  \vdash P
cases h with hP hQ,
  hP : P,
 hQ : Q
  \vdash P
exact hP,
 no goals
-/
-- Comentarios:
-- + La táctica (cases h with h1 h2), cuando la hipótesis h es una
     conjunción aplica la regla de eliminación de la conjunción; es
     decir, si h es (P κ Q), entonces elimina h y añade las hipótesis
     (h1 : P) y (h2 : Q).
-- 2ª demostración (con rintro y exact)
example : P \land Q \rightarrow P :=
begin
  rintro (hP, hQ),
  exact hP,
end
-- Prueba:
 P Q : Prop
 \vdash P \land 0 \rightarrow P
rintro (hP, hQ),
 hP:P,
 hQ : Q
  \vdash P
exact hP,
 no goals
-/
-- Comentarios:
-- + La táctica (rintro (h1, h2)), cuando la conclusión es una
     implicación cuyo antecedente es una conjunción, aplica las reglsa
     de introducción de la implicación y de eliminación de la conjunción;
```

```
es decir, si la conclusión es (P ∧ Q → R) entonces añade las
    hipótesis (h1 : P) y (h2 : Q) y cambia la conclusión a R.
-- 3ª demostración (con rintro y assumption)
example : P \land Q \rightarrow P :=
begin
 rintro (hP, hQ),
 assumption,
end
-- Comentarios:
-- + la táctica assumption concluye la demostración si la conclusión
    coincide con alguna de las hipótesis.
-- 4ª demostración (estructurada)
example : P \land Q \rightarrow P :=
begin
 assume h : P \land Q,
 show P, from h.1,
end
-- 5ª demostración (estructurada)
example : P \land Q \rightarrow P :=
assume h, h.1
-- 6ª demostración (con término de prueba)
example : P \land Q \rightarrow P :=
\lambda (hP,_), hP
-- 7º demostración (con lema)
example : P \land Q \rightarrow P :=
and.left
-- Comentarios:
-- + Se usa el lema
```

#### 3.3.2. Introducción de la conjunción

```
begin
  split,
 { exact hP },
  { apply hPQ,
    exact hP },
end
-- Comentario
-- La táctica split, cuando la conclusión es una conjunción, aplica la
-- regla de eliminación de la conjunción; es decir, si la conclusión es
-- (P Λ Q), entonces crea dos subojetivos: el primero en el que la
-- conclusión es P y el segundo donde es Q.
-- 2ª demostración
-- ==========
example
 (hP : P)
 (hPQ : P \rightarrow Q)
  : P ^ Q :=
begin
 split,
 { exact hP },
 { exact hPQ hP },
end
-- 3ª demostración
-- ===========
example
 (hP : P)
  (hPQ : P \rightarrow Q)
 : P ^ Q :=
begin
 have hQ : Q := hPQ hP,
 show P \wedge Q, by exact \langle hP, hQ \rangle,
end
-- 4ª demostración
-- ===========
example
 (hP : P)
```

```
(hPQ : P \rightarrow Q)
  : P ^ Q :=
begin
  show P \( \text{Q} \), by exact \( \text{hP, hPQ hP} \),
end
-- 4ª demostración
-- ==========
example
 (hP : P)
  (hPQ : P \rightarrow Q)
  : P ^ Q :=
begin
  exact (hP, hPQ hP),
end
-- 5ª demostración
-- ===========
example
 (hP : P)
  (hPQ : P \rightarrow Q)
  : P ^ Q :=
by exact (hP, hPQ hP)
-- 6ª demostración
-- ==========
example
 (hP : P)
 (hPQ : P \rightarrow Q)
 : P ^ Q :=
(hP, hPQ hP)
-- 7ª demostración
-- ==========
example
 (hP : P)
  (hPQ : P \rightarrow Q)
  : P ^ Q :=
and.intro hP (hPQ hP)
-- Comentario: Se ha usado el lema
```

```
-- + and.intro : P \rightarrow Q \rightarrow P \wedge Q

-- 8^{\underline{a}} demostración
-- ============

example
(hP : P)
(hPQ : P \rightarrow Q)
: P \lambda Q :=
by tauto
```

## 3.4. Reglas de la disyunción

## 3.4.1. Eliminación de la disyunción

```
(hQR : Q \rightarrow R)
  : P \lor Q \rightarrow R :=
begin
 intro h,
  cases h with hP hQ,
 { exact hPR hP },
  { exact hQR hQ },
end
-- Comentario
-- La táctica (cases h with h1 h2), cuando la hipótesis h es una
-- disyunción aplica la regla de eliminación de la disyunción; es decir,
-- si h es (P v Q), entonces elimina h y crea dos casos: uno añadiendo
-- la hipótesis (h1 : P) y otro añadiendo la hipótesis (h2 : Q).
-- 2ª demostración
-- ==========
example
  (hPR : P \rightarrow R)
  (hQR : Q \rightarrow R)
  : P \lor Q \rightarrow R :=
begin
  rintro (hP | hQ),
  { exact hPR hP },
  { exact hQR hQ },
end
-- Comentario
-- La táctica (rintro (h1 | h2)), cuando la conclusión es una
-- implicación cuyo antecedente es una disyunción, aplica las regla des
-- introducción de la implicación y de eliminación de la disyunción; es
-- decir, si la conclusión es (P v Q → R) entonces crea dos casos: en el
-- primero añade la hipótesis (h1 : P) y cambia a conclusión a R; en el
-- segundo añade la hipótesis (h2 : Q) y cambia la conclusión a R.
-- 3ª demostración
-- ===========
example
 (hPR : P \rightarrow R)
```

```
(hQR : Q \rightarrow R)
  : P ∨ Q → R :=
λ h, or.elim h hPR hQR
-- Comentario: Se ha usado el lema
-- + or.elim : P \lor Q \rightarrow (P \rightarrow R) \rightarrow (Q \rightarrow R) \rightarrow R
-- 3ª demostración
-- ==========
example
  (hPR : P \rightarrow R)
  (hQR : Q \rightarrow R)
  : P V Q \rightarrow R :=
or.rec hPR hQR
-- Comentario: Se ha usado el lema
-- + or.rec : (P \rightarrow R) \rightarrow (Q \rightarrow R) \rightarrow P \lor Q \rightarrow R
-- 4ª demostración
-- ==========
example
  (hPR : P \rightarrow R)
  (hQR : Q \rightarrow R)
  : P V Q \rightarrow R :=
by tauto
```

# 3.5. Ejercicios

### 3.5.1. Monotonía de la suma por la izquierda

```
-- Ejercicio. Demostrar que si a, b y c son números reles tales que -- a \le b, entonces c + a \le c + b. -- Indicación: Se puede usar el lema -- sub_nonneg : 0 \le a - b \Leftrightarrow b \le a
```

```
import data.real.basic
variables {a b c : R}
-- 1ª demostración
-- ==========
example
 (hab : a \leq b)
  : c + a \le c + b :=
begin
  rw ← sub_nonneg,
 have h : (c + b) - (c + a) = b - a,
 { ring, },
  { rw h,
   rw sub_nonneg,
    exact hab, },
end
-- Comentario: Se ha usado el lema
-- + sub nonneg : 0 \le a - b ⇔ b \le a
-- 2ª demostración
-- ==========
example
 (hab : a \leq b)
 : c + a \le c + b :=
begin
 rw ← sub_nonneg,
 calc 0 \le b - a : by exact sub_nonneg.mpr hab
      \dots = c + b - (c + a) : by exact (add_sub_add_left_eq_sub b a c).symm,
end
-- Comentario: Se usa el lema
-- + add\_sub\_add\_left\_eq\_sub : c + a - (c + b) = a - b
-- 3ª demostración
-- ==========
example
 (hab : a \leq b)
 : c + a \le c + b :=
begin
 rw ← sub_nonneg,
 calc 0 \leq b - a
                             : sub nonneg.mpr hab
```

```
\dots = c + b - (c + a) : (add_sub_add_left_eq_sub_b a c).symm,
end
-- 4ª demostración
-- ==========
example
 (hab : a \leq b)
 : c + a \le c + b :=
begin
 rw <mark>←</mark> sub_nonneg,
 calc 0 ≤ b - a : sub_nonneg.mpr hab
     ... = c + b - (c + a) : ring
end
-- 5ª demostración
-- ==========
example
 (hab : a \leq b)
 : c + a \le c + b :=
begin
 rw ← sub_nonneg,
 simp,
 exact hab,
end
-- 6ª demostración
-- ===========
example
 (hab : a ≤ b)
 : c + a \le c + b :=
 rw ← sub_nonneg,
 simp [hab],
end
-- Comentario:
-- + La táctica (simp [h]) aplica reglas de simplificación, ampliadas con
-- h, a la conclusión.
-- 7º demostración
-- ==========
```

```
example
 (hab : a \leq b)
  : c + a \le c + b :=
begin
 simp [hab],
end
-- 8ª demostración
-- ==========
example
 (hab : a \leq b)
 : c + a \le c + b :=
by simp [hab]
-- 9ª demostración
-- ===========
example
 (hab : a \leq b)
 : c + a \le c + b :=
add_le_add_left hab c
-- Comentario: Se ha usado el lema
-- + add_le_add_left: a \le b \rightarrow \forall (c: \mathbb{R}), c + a \le c + b
-- 10ª demostración
-- ==========
example
  (hab : a \leq b)
  : c + a \le c + b :=
by linarith
-- 11ª demostración
-- ==========
example
 (hab : a \leq b)
 : c + a \le c + b :=
by finish
```

#### 3.5.2. Monotonía de la suma por la derecha

```
-- Ejercicio. Demostrar que si a, b y c son números reles tales que
-- a ≤ b, entonces a + c ≤ b + c.
import data.real.basic
variables {a b c : ℝ}
-- 1ª demostración
-- ==========
example
 (hab : a \leq b)
 : a + c \le b + c :=
  rw ← sub_nonneg,
 have h : (b + c) - (a + c) = b - a,
 { ring, },
 { rw h,
   rw sub nonneg,
   exact hab, },
end
-- 2ª demostración
-- ==========
example
 (hab : a \leq b)
  : a + c \le b + c :=
begin
 rw ← sub_nonneg,
  calc 0 ≤ b - a : by exact sub_nonneg.mpr hab
      \dots = b + c - (a + c) : by exact (add sub add right eq sub b a c).symm,
end
-- Comentario: Se usa el lema
-- + add\_sub\_add\_right\_eq\_sub : a + c - (b + c) = a - b
-- 3ª demostración
-- ==========
```

```
example
 (hab : a \leq b)
 : a + c \le b + c :=
begin
 rw ← sub_nonneg,
                    : sub_nonneg.mpr hab
 calc 0 ≤ b - a
      \dots = b + c - (a + c) : (add_sub_add_right_eq_sub_b a c).symm,
end
-- 4ª demostración
-- ==========
example
 (hab : a \leq b)
  : a + c \le b + c :=
begin
  rw ← sub_nonneg,
 calc 0 ≤ b - a : sub_nonneg.mpr hab
      ... = b + c - (a + c) : by ring,
end
-- 5ª demostración
-- ===========
example
 (hab : a \leq b)
 : a + c \le b + c :=
begin
 rw ← sub_nonneg,
 simp,
 exact hab,
end
-- 6ª demostración
-- ==========
example
 (hab : a \leq b)
 : a + c \le b + c :=
begin
 rw ← sub_nonneg,
 simp [hab],
end
-- 7ª demostración
```

```
-- ==========
example
 (hab : a ≤ b)
 : a + c \le b + c :=
begin
 simp [hab],
end
-- 8ª demostración
-- ==========
example
 (hab : a \leq b)
 : a + c \le b + c :=
by simp [hab]
-- 9ª demostración
-- ============
example
 (hab : a \leq b)
 : a + c \le b + c :=
add_le_add_right hab c
-- Comentario: Se ha usado el lema
-- + add_{e_add_right} : a \le b \rightarrow \forall (c : \mathbb{R}), a + c \le b + c
-- 10ª demostración
-- ==========
example
 (hab : a \leq b)
 : a + c \le b + c :=
by linarith
-- 11ª demostración
-- ==========
example
 (hab : a \leq b)
 : a + c \le b + c :=
by finish
```

#### 3.5.3. La suma de no negativos es expansiva

```
-- Ejercicio 1.. Demostrar si a y b son números reales y a es no
-- negativo, entonces b \le a + b
import data.real.basic
variables {a b : ℝ}
-- 1ª demostración
-- ==========
example
 (ha : 0 \le a)
 : b \le a + b :=
 calc b = 0 + b : by rw zero add
     ... ≤ a + b : by exact add_le_add_right ha b,
-- Comentario: Se ha usado el lema
-- + zero \ add : 0 + a = a
-- 2ª demostración
-- ==========
example
 (ha : 0 \le a)
 : b \le a + b :=
begin
 calc b = 0 + b : (zero\_add b).symm
    ... ≤ a + b : add_le_add_right ha b,
end
-- 3ª demostración
-- ==========
example
 (ha : 0 \le a)
 : b \le a + b :=
begin
 calc b = 0 + b : by ring
```

```
... ≤ a + b : by exact add_le_add_right ha b,
end
-- 4ª demostración
example
 (ha : 0 \le a)
 : b \le a + b :=
by simp [ha]
-- 5ª demostración
-- ==========
example
 (ha : 0 \le a)
  : b ≤ a + b :=
by linarith
-- 6ª demostración
-- ==========
example
 (ha : 0 \le a)
  : b ≤ a + b :=
by finish
-- 7ª demostración
-- ==========
example
 (ha : 0 \le a)
 : b ≤ a + b :=
le_add_of_nonneg_left ha
-- Comentario: Se ha usado el lema
-- + le add of nonneg left : 0 \le b \rightarrow a \le b + a
-- Ejercicio 2. Demostrar si a y b son números reales y b es no
-- negativo, entonces a ≤ a + b
-- 1ª demostración
-- ==========
```

```
example
  (\mathsf{h}\mathsf{b} : 0 \leq \mathsf{b})
  : a ≤ a + b :=
begin
 calc a = a + 0: by rw add zero
     ... ≤ a + b : by exact add le add left hb a,
end
-- Comentario: Se ha usado el lema
-- + add_le_add_left: a \le b \rightarrow \forall (c: \mathbb{R}), c + a \le c + b
-- 2ª demostración
-- ==========
example
  (hb : 0 \le b)
  : a ≤ a + b :=
begin
  calc a = a + 0: (add zero a).symm
    ... ≤ a + b : add_le_add_left hb a,
end
-- 3ª demostración
-- ===========
example
 (\mathsf{h}\mathsf{b} : 0 \le \mathsf{b})
 : a ≤ a + b :=
begin
  calc a = a + 0: by ring
     ... ≤ a + b : add le add left hb a,
end
-- 4ª demostración
-- ===========
example
 (hb : 0 \le b)
  : a ≤ a + b :=
by simp [hb]
-- 5ª demostración
-- ===========
```

```
example
  (hb : 0 \le b)
  : a ≤ a + b :=
by linarith
-- 6ª demostración
-- ===========
example
 (\mathsf{h}\mathsf{b} : 0 \leq \mathsf{b})
 : a ≤ a + b :=
by finish
-- 7º demostración
-- ==========
example
 (\mathsf{h}\mathsf{b} : 0 \le \mathsf{b})
 : a ≤ a + b :=
le add of nonneg right hb
-- Comentario: Se usa el lema
-- + le_add_of_nonneg_right : 0 \le b \rightarrow a \le a + b
```

## 3.5.4. Suma de no negativos

```
: 0 ≤ a + b :=
begin
  calc 0 \le a : ha
     \dots \le a + b: le add of nonneg right hb,
end
-- 2ª demostración
-- ==========
example
  (ha : 0 \le a)
  (hb : 0 \le b)
  : 0 \le a + b :=
add_nonneg ha hb
-- Comentario: Se usa el lema
-- + add_nonneg : 0 \le a \to 0 \le b \to 0 \le a + b
-- 3ª demostración
-- ==========
example
  (ha : 0 \le a)
  (hb : 0 \le b)
 : 0 ≤ a + b :=
by linarith
```

### 3.5.5. Suma de desigualdades

```
example
  (hab : a \leq b)
  (hcd : c \leq d)
  : a + c \le b + d :=
begin
    a + c \le b + c : add_le_add_right hab c
    ... ≤ b + d : add le add left hcd b,
end
-- 2ª demostración
example
  (hab : a \leq b)
  (hcd : c \leq d)
  : a + c \le b + d :=
begin
  have h1 : a + c \le b + c :=
    add le add right hab c,
  have h2 : b + c \le b + d :=
    add le add left hcd b,
  show a + c \le b + d,
    from le_trans h1 h2,
end
-- Comentario: Se ha usado el lema
-- + le_trans: a \le b \rightarrow b \le c \rightarrow a \le c
-- 3ª demostración
-- =========
example
  (hab : a \leq b)
  (hcd : c \leq d)
  : a + c \le b + d :=
add_le_add hab hcd
-- Comentario: Se ha usado el lema
--+ add\_le\_add: a \le b \rightarrow c \le d \rightarrow a + c \le b + d
-- 4ª demostración
-- =========
example
  (hab : a \leq b)
  (hcd : c \leq d)
```

```
: a + c ≤ b + d :=
by linarith
```

### 3.5.6. Monotonía de la multiplicación por no negativo

```
-- Ejercicio. Demostrar que si a, b y c son números reales tales que
-- 0 \le c \ y \ a \le b, entonces a*c \le b*c.
import data.real.basic
variables {a b c : R}
-- 1ª demostración
-- ==========
example
 (hc : 0 \le c)
  (hab : a \leq b)
  : a * c \le b * c :=
begin
  rw ← sub_nonneg,
  have h : b * c - a * c = (b - a) * c,
  { ring },
  { rw h,
    apply mul_nonneg,
    { rw sub_nonneg,
      exact hab },
    { exact hc }},
end
-- Comentario: Se ha usado el lema
-- + mul nonneg : 0 \le a \to 0 \le b \to 0 \le a * b
-- 2ª demostración
- - ===========
example
 (hc : 0 \le c)
 (hab : a \leq b)
```

```
: a * c ≤ b * c :=
begin
  have hab' : 0 \le b - a,
  { rw ← sub nonneg at hab,
    exact hab, },
  have h1 : 0 \le (b - a) * c,
  { exact mul nonneg hab' hc, },
  have h2 : (b - a) * c = b * c - a * c,
  { ring, },
  have h3 : 0 \le b * c - a * c,
  { rw h2 at h1,
    exact h1, },
  rw sub nonneg at h3,
  exact h3,
end
-- 3ª demostración
-- ==========
example
  (hc : 0 \le c)
  (hab : a \leq b)
  : a * c \le b * c :=
  have hab' : 0 \le b - a,
  { rwa ← sub_nonneg at hab, },
  have h1 : 0 \le (b - a) * c,
  { exact mul_nonneg hab' hc },
  have h2 : (b - a) * c = b * c - a * c,
  { ring, },
 have h3 : 0 \le b * c - a * c,
 { rwa h2 at h1, },
  rwa sub nonneg at h3,
end
-- Comentario:
-- + La táctica (rwa h at h'), cuando h es una igualdad. sustituye en la
    hipótesis h' el término izquierdo de h por el derecho y, a
    continuación, aplica assumption.
-- + La táctica (rwa ← h at h'), cuando h es una igualdad, sustituye en
    la hipótesis h' el término derecho de h por el izquierdo y, a
     continuación, aplica assumption.
-- 4ª demostración
-- ==========
```

```
example
 (hc : 0 \le c)
  (hab : a \leq b)
 : a * c ≤ b * c :=
mul mono nonneg hc hab
-- Comentario: Se usa el lema
-- + mul mono nonneg : 0 \le c \rightarrow a \le b \rightarrow a * c \le b * c
-- 5ª demostración
-- ==========
example
 (hc : 0 \le c)
  (hab : a \leq b)
  : a * c ≤ b * c :=
by nlinarith
-- Comentario:
-- + La táctica nlinarith es una extensión de linarith con un
-- preprocesamiento que permite resolver problemas aritméticos no
-- lineales.
```

# Capítulo 4

# **Apéndices**

#### 4.1. Resumen de tácticas usadas

- (apply h), cuando h es una implicación, aplica la regla de eliminación de la implicación; es decir, si h es (P → Q) y la conclusión coincide con Q, entonces sustituye la conclusión por P.
- assumption concluye la demostración si la conclusión coincide con alguna de las hipótesis.
- exact h concluye la demostración si h es del tipo de la conclusión.
- (cases h with h1 h2), cuando la hipótesis h es una equivalencia aplica la regla de eliminación de la equivalencia; es decir, si h es (P ↔ Q), entonces elimina h y añade las hipótesis (h1 : P → Q) y (h2 : Q → P).
- (cases h with h1 h2), cuando la hipótesis h es una conjunción aplica la regla de eliminación de la conjunción; es decir, si h es (P λ Q), entonces elimina h y añade las hipótesis (h1 : P) y (h2 : Q).
- (cases h with h1 h2), cuando la hipótesis h es una disyunción aplica la regla de eliminación de la disyunción; es decir, si h es (P v Q), entonces elimina h y crea dos casos: uno añadiendo la hipótesis (h1 : P) y otro añadiendo la hipótesis (h2 : Q).
- intro h, cuando la conclusión es una implicación, aplica la regla de introducción de la implicación; es decir, si la conclusión es (P → Q) entonces añade la hipótesis (h : P) y cambia la conclusión a Q.
- finish demuestra la conclusión de forma automática.
- linarith demuestra la conclusión mediante aritmética lineal.

- nlinarith es una extensión de linarith con un preprocesamiento que permite resolver problemas aritméticos no lineales.
- ring demuestra la conclusión normalizando las expresiones con las regñlas de los anillos.
- rintro (h1, h2), cuando la conclusión es una implicación cuyo antecedente es una conjunción, aplica las regla de introducción de la implicación y de eliminación de la conjunción; es decir, si la conclusión es (P ∧ Q → R) entonces añade las hipótesis (h1 : P) y (h2 : Q) y cambia la conclusión a R.
- rintro (h1 | h2), cuando la conclusión es una implicación cuyo antecedente es una disyunción, aplica las regla des introducción de la implicación y de eliminación de la disyunción; es decir, si la conclusión es (P v Q → R) entonces crea dos casos: en el primero añade la hipótesis (h1 : P) y cambia a conclusión a R; en el segundo añade la hipótesis (h2 : Q) y cambia la conclusión a R.
- rw h cuando h es una igualdad sustituye en la conclusión el término izquierdo de h por el derecho.
- rw h, cuando h es una equivalencia como (P ↔ Q), sustituye en la conclusión P por Q.
- rw ← h cuando h es una igualdad sustituye en la conclusión el término derecho de h por el izquierdo
- rw h at h' cuando h es una igualdad sustituye en la hipótesis h' el término izquierdo de h por el derecho.
- rw h at h' cuando h es una equivalencia como (P ↔ Q) sustituye en la hipótesis h' la fórmula P por Q.
- rw ← h at h' cuando h es una igualdad sustituye en la hipótesis h' el término derecho de h por el izquierdo
- rw ← h at h' cuando h es una equivalencia como (P ↔ Q) sustituye en la hipótesis h' la fórmula Q por P.
- rwa h cuando h es una igualdad sustituye en la conclusión el término izquierdo de h por el derecho y, a continuación, aplica assumption.
- rwa h at h' cuando h es una igualdad sustituye en la hipótesis h' el término izquierdo de h por el derecho y, a continuación, aplica assumption..

- rwa ← h at h' cuando h es una igualdad sustituye en la hipótesis h' el término derecho de h por el izquierdo y, a continuación, aplica assumption.
- simp aplica reglas de simplificación a la conclusión.
- simp [h] aplica reglas de simplificación, ampliadas con h, a la conclusión.
- solve\_by\_elim intenta demostrar el objetivo aplicándole reglas de eliminación.
- split, cuando la conclusión es una conjunción, aplica la regla de eliminación de la conjunción; es decir, si la conclusión es (P Λ Q), entonces crea dos subojetivos: el primero en el que la conclusión es P y el segundo donde es Q.
- tauto demuestra automáticamente las tautologías.

#### 4.1.1. Demostraciones estructuradas

- (assume h : P), cuando la conclusión es de la forma (P → Q), añade la hipótesis P y cambia la conclusión a Q.
- (have h : e) genera dos subojetivos: el primero tiene como conclusión e y el segundo tiene la conclusión actual pero se le añade la hipótesis (h : e).
- show P, from h demuestra la conclusión con la prueba h.

### 4.1.2. Composiciones y descomposiciones

- Si h1 es una demostración de (P → Q) y h2 es una demostración de P, entonces (h1 h2) es una demostración de Q.
- Si h es la conjunción (P Λ Q), entonces h.1 es P y h.2 es Q.
- Si h es la equivalencia (P  $\leftrightarrow$  Q), entonces h.1 es (P  $\rightarrow$  Q) y h.2 es (Q  $\rightarrow$  P).
- Si h es una igualdad entonces h ➤ h' es la expresión obtenida sustituyendo en h' el término izquierdo de h por el derecho.

#### 4.2. Resumen de teoremas usados

Los teoremas utilizados son los siguientes:

```
+ add assoc : (a + b) + c = a + (b + c)
+ add_{comm} : a + b = b + a
+ add_le_add : a \le b \rightarrow c \le d \rightarrow a + c \le b + d
+ add le add left : a \le b \rightarrow \forall (c : \mathbb{R}), c + a \le c + b
+ add le add right : a \le b \rightarrow \forall (c : \mathbb{R}), a + c \le b + c
+ add mul : (a + b) * c = a * c + b * c
+ add nonneg : 0 \le a \to 0 \le b \to 0 \le a + b
+ add_sub : a + (b - c) = (a + b) - c
+ add sub add left eq sub : c + a - (c + b) = a - b
+ add sub add right eq sub : a + c - (b + c) = a - b
+ add zero : a + 0 = a
+ and.intro : P \rightarrow Q \rightarrow P \wedge Q
+ and.left : P ∧ Q → P
+ eq.trans : a = b \rightarrow b = c \rightarrow a = c
+ id : P \rightarrow P
+ le_add_of_nonneg_left : 0 ≤ b → a ≤ b + a
+ le add of nonneg right : 0 \le b \rightarrow a \le a + b
+ le trans: a \le b \rightarrow b \le c \rightarrow a \le c
+ \text{ mul assoc} : (a * b) * c = a * (b * c)
+ mul\_comm : a * b = b * a
+ mul mono nonneg : 0 \le c \rightarrow a \le b \rightarrow a * c \le b * c
+ mul nonneg : 0 \le a \to 0 \le b \to 0 \le a * b
+ mul_sub : a * (b - c) = a * b - a * c
+ or.elim : P \lor Q \rightarrow (P \rightarrow R) \rightarrow (Q \rightarrow R) \rightarrow R
+ or.rec : (P \rightarrow R) \rightarrow (Q \rightarrow R) \rightarrow P \lor Q \rightarrow R
+ pow_two : a^2 = a * a
+ sub_nonneg : 0 \le a - b \Leftrightarrow b \le a
+ sub_self : a - a = 0
+ sub_sub : (a - b) - c = a - (b + c)
+ two_mul : 2 * a = a + a
+ zero_add : 0 + a = a
```

## 4.3. Estilos de demostración

Demostración	Demostración	Término de prueba
con tácticas	estructurada	en bruto
intro x,	fix x,	λ x,
intro h,	assume h,	λh,
have $k := \_$ ,	have $k := \_$ ,	have k := _,
let x := _,	let $x := _in$	let x := _ in
exact (_ : P)	show P, from _	_:P

- intro x equivale a λ x, \_
- apply f equivale a f \_ \_ \_
- refine e1 (e2 \_) (e3 \_) equivale a e1 (e2 \_) (e3 \_)
- exact e equivale a e
- change e equivale a (\_ : e)
- rw h equivale a eq.rec\_on h \_
- induction e equivale a T.rec\_on foo \_ \_ donde T es el tipo de e
- cases e equivale a T.cases\_on e \_ \_ donde T es el tipo de e
- split equivale a and.intro \_ \_
- have  $x : t = equivale a (\lambda x, _) t$
- let x : t = equivale a let x : t in <math>=
- revert x equivale a \_ x